Automated One-Loop Computation in Quarkonium Process Within NRQCD Framework

Feng Feng

Center for High Energy Physics, PEKING University

15th International Workshop on Advanced Computing and Analysis Techniques in Physics

May 16 — 21, 2013 Beijing, CHINA

Outline

- General steps for the one-loop computation in quarkonium physics within the NRQCD framework and basic tools used in each step
  - Generating Feynman diagrams and amplitudes
    FeynArts, QGraf ···
  - Dirac- and Color-algebra simplification
    FeynCalc, FORM, FormLinkFeynCalc ···
  - Partial fraction and IBP reduction to Master Integrals (MI)
    $Apart, FIRE ···
  - Post-processing final results

- Specific process in the examples

$$e^+ + e^- \to \gamma^* \to J/\psi + \text{S-, P- or D-wave}$$

- Summary

Generating Feynman Diagrams and Amplitudes

- Generating Feynman diagrams
  - Replacing hadrons with on-shell partonic states

$$J/\psi \to c(p_1)\bar{c}(p_2) \qquad J/\psi \to c(p_1)\bar{c}(p_2)g(k)$$

  - FeynArts – Mathematica 9
    http://www.feynarts.com
  - QGraf – Fortran 77
    http://cfif.ist.utl.pt/~paulo/qgraf.html
- Generating Amplitude for each diagram
  - Standard Model
  - Adjusting for later processing
  - Export to files

Generating Feynman Diagrams and Amplitudes

```
<< FeynArts`FeynArts`

$ExcludeTopologies[V4onExt] = FreeQ[Cases[#, Propagator[External][__]], Vertex[4]] &;

top = CreateTopologies[1, 1 → 4, ExcludeTopologies → {WFCorrections, Tadpoles, V4onExt}];

tmp = InsertFields[top, {V[1]} → {F[3], -F[3], F[3], -F[3]}, Model → "SMQCD",
   ExcludeParticles → {V[1], V[2], V[3], V[4], S[_], F[4]}, InsertionLevel → {Classes}];

tmp1 =
  DiagramSelect[tmp,
   (FreeQ[List @@ #2, {___, PatternSequence[Propagator[Outgoing][_, v_, Field[2]], ___,
         Propagator[Outgoing][_, v_, Field[3]]], ___}]) &];
tmp2 =
  DiagramSelect[tmp1,
   (FreeQ[List @@ #2, {___, PatternSequence[Propagator[Outgoing][_, v_, Field[4]], ___,
         Propagator[Outgoing][_, v_, Field[5]]], ___}]) &];

all = tmp2;

all = DiagramDelete[all, 3 ... 4, 13 ... 14, 25 ... 26, 33 ... 34, 42 ... 43];

Paint[all, ColumnsXRows → {10, 7}];
```



$$e^+ + e^- \to \gamma^* \to J/\psi + \eta_c$$

Generating Feynman Diagrams and Amplitudes

```
FAToFC[d__] := Module[{temp},
  temp = CreateFeynAmp[d, PreFactor → 1] /. {FourMomentum[Incoming, 1] → p3 + p4, FourMomentum[Outgoing, 1] → p3/2 + q3,
     FourMomentum[Outgoing, 2] → p3/2 - q3, FourMomentum[Outgoing, 3] → p4/2 + q4, FourMomentum[Outgoing, 4] → p4/2 - q4,
     MQU[Index[g_, i_]] → mc, MQD[Index[g_, i_]] → mc, EL → e, GS → Gstrong};
  temp = (Part[#, 3] &) /@ (List @@ ToFA1Conventions[temp]);
  temp = temp /. {DiracMatrix[li_].ChiralityProjector[1] ↦ DiracMatrix[li] / 2,
     DiracMatrix[li_].ChiralityProjector[-1] ↦ DiracMatrix[li] / 2};
  temp = temp /. {exp_DiracTrace ↦ FactorSUNT[exp]};
  temp = temp /. {IndexDelta[Index[x1_, y1_], Index[x2_, y2_]] ↦ 1, SUNT[c_, _, _] ↦ SUNT[c], SumOver[__] → 1,
     PolarizationVector[___] → 1};
  Return[temp];
 ];

Amp = FAToFC[all];

Amp = Amp /. fm : DiracSpinor[p4/2 + q4, mc].(mx___ /; FreeQ[{mx}, DiracSpinor]).DiracSpinor[q3 - p3/2, mc] ↦ $LineA1[Dot[mx]];
Amp = Amp /. fm : DiracSpinor[p3/2 + q3, mc].(mx___ /; FreeQ[{mx}, DiracSpinor]).DiracSpinor[q4 - p4/2, mc] ↦ $LineA2[Dot[mx]];

Amp = Amp /. fm : DiracSpinor[p4/2 + q4, mc].(mx___ /; FreeQ[{mx}, DiracSpinor]).DiracSpinor[q4 - p4/2, mc] ↦ $LineB1[Dot[mx]];
Amp = Amp /. fm : DiracSpinor[p3/2 + q3, mc].(mx___ /; FreeQ[{mx}, DiracSpinor]).DiracSpinor[q3 - p3/2, mc] ↦ $LineB2[Dot[mx]];

SetDirectory[ToFileName[NotebookDirectory[], "dmp"]];
Export["Amp.m", Amp];
ResetDirectory[];
```

$$e^+ + e^- \to \gamma^* \to J/\psi + \eta_c$$

Dirac- and Color-Algebra Simplification

- Covariant spin projectors   Bodwin, Petrelli (2002) & Braatten, Lee (2003)

$$v(\bar{p})\bar{u}(p) \rightarrow \frac{1}{4\sqrt{2}E(E+m_c)}(\not{\bar{p}} - m_c)\left[\gamma_5, \not{\epsilon}^*\right](\not{P} + 2E)(\not{p} + m_c) \otimes \left[\frac{\mathbf{1}}{\sqrt{N_c}}, T^a\right]$$

- Performing Dirac- and Color-algebra
  - FORM – written in C
    http://www.nikhef.nl/~form/
  - FeynCalc – Mathematica 9
    http://www.feyncalc.org/
  - FeynCalcFormLink – Combine FORM & FeynCalc
    http://www.feyncalc.org/formlink/

FormCalc : Exchange data through Input & Output files

FormLink : Exchange data through Piping

Introduction to FeynCalcFormLink

- Communication Between Processes
  - Input & Output Files
  - Piping

FormCalc



FeynCalcFormLink



Introduction to FeynCalcFormLink

- Basic ideas of FormLink
  - FormLink : create two unnamed pipes: r#, w#
  - FormLink : form -pipe r#, w# init
  - PID in w# : FORM → FormLink
  - PID,PID in r# : FORM ← FormLink
  - Form start running init.frm
- Content of init.frm

```
Off Statistics;
#ifndef 'PIPES_'
#message "No pipes found";
.end;
#endif
#if ('PIPES_' <= 0)
#message "No pipes found";
.end;
#endif
```

```
#procedure put(fmt, mexp)
#toexternal 'fmt', 'mexp'
#toexternal "#THE-END-MARK#"
#endprocedure
#setexternal 'PIPE1_';
#toexternal "OK"
#fromexternal
.end
```

Introduction to FeynCalcFormLink

- Interactive FORM

```
In[1]:= << FormLink`

In[2]:= FormStart[];

In[3]:= FormWrite["Symbol a,b,c;"];

In[4]:= FormWrite["L exp=(a+b+c)^3;"];

In[5]:= FormWrite[".sort;"];

In[6]:= FormWrite["#call put(\"%E\", exp);"];

In[7]:= FormWrite[".sort;"];

In[8]:= TraditionalForm@ToExpression@FormRead[]

Out[8]//TraditionalForm=
```

$$a^3+3a^2b+3a^2c+3ab^2+6abc+3ac^2+b^3+3b^2c+3bc^2+c^3$$

```
In[9]:= FormWrite["L exp2 = exp^2;"];

In[10]:= FormWrite[".sort;"];

In[11]:= FormWrite["#call put(\"%E\", exp2);"];

In[12]:= FormWrite[".sort;"];

In[13]:= TraditionalForm@ToExpression@FormRead[]

Out[13]//TraditionalForm=
```

$$a^6+6a^5b+6a^5c+15a^4b^2+30a^4bc+15a^4c^2+20a^3b^3+60a^3b^2c+60a^3bc^2+20a^3c^3+15a^2b^4+60a^2b^3c+90a^2b^2c^2+60a^2bc^3+15a^2c^4+6ab^5+30ab^4c+60ab^3c^2+60ab^2c^3+30abc^4+6ac^5+b^6+6b^5c+15b^4c^2+20b^3c^3+15b^2c^4+6bc^5+c^6$$

Introduction to FeynCalcFormLink

- A Simple Trace with FeynCalc

```
In[1]:= << HighEnergyPhysics`fc`

In[2]:= Tr[GS[p1, p2, p3, p4, p5, p6]]
```

Out[2]= $4\,(p1\cdot p6\,p2\cdot p5\,p3\cdot p4 - p1\cdot p5\,p2\cdot p6\,p3\cdot p4 + p1\cdot p2\,p3\cdot p4\,p5\cdot p6 - p1\cdot p6\,p2\cdot p4\,p3\cdot p5 + p1\cdot p4\,p2\cdot p6\,p3\cdot p5 + p1\cdot p5\,p2\cdot p4\,p3\cdot p6 - p1\cdot p4\,p2\cdot p5\,p3\cdot p6 + p1\cdot p6\,p2\cdot p3\,p4\cdot p5 - p1\cdot p3\,p2\cdot p6\,p4\cdot p5 + p1\cdot p2\,p3\cdot p6\,p4\cdot p5 - p1\cdot p5\,p2\cdot p3\,p4\cdot p6 + p1\cdot p3\,p2\cdot p5\,p4\cdot p6 - p1\cdot p2\,p3\cdot p5\,p4\cdot p6 + p1\cdot p4\,p2\cdot p3\,p5\cdot p6 - p1\cdot p3\,p2\cdot p4\,p5\cdot p6)$

- Do it with FeynCalcFormLink

```
In[8]:= ? FormLink
```

FormLink[" form statements "] runs the form statements in Form and returns the result to Mathematica.
If only one Local assignment is present the return value is that result. If more Form Local variables are present, a list of results is returned.

```
In[9]:= ? FeynCalcFormLink
```

FeynCalcFormLink[expr] translates the FeynCalc expression expr to FORM, calculates it, pipes it back to Mathematica and translates it to FeynCalc syntax.

Introduction to FeynCalcFormLink

- Do it with FeynCalcFormLink

```
In[1]:= << FeynCalcFormLink`

In[2]:= exp = DiracTrace[GS[p1, p2, p3, p4, p5, p6]]
```

Out[2]= $\mathrm{tr}((\gamma\cdot p1).(\gamma\cdot p2).(\gamma\cdot p3).(\gamma\cdot p4).(\gamma\cdot p5).(\gamma\cdot p6))$

```
In[3]:= FeynCalcFormLink[exp]
```

```
Vectors p1,p2,p3,p4,p5,p6;
Format Mathematica;
L resFL = (g_(1,p1)*g_(1,p2)*g_(1,p3)*g_(1,p4)*g_(1,p5)*g_(1,p6));
trace4,1;
contract 0;
.sort;
#call put("%E", resFL)
#fromexternal
```

Piping the script to FORM and running FORM

Time needed by FORM : 0.016 seconds. FORM finished. Got the result back to Mathematica as a string.

Start translation to Mathematica / FeynCalc syntax

Total wall clock time used: 1.31 seconds. Translation to Mathematica and FeynCalc finished.

Out[3]= $4\,p1\cdot p6\,p2\cdot p5\,p3\cdot p4 - 4\,p1\cdot p5\,p2\cdot p6\,p3\cdot p4 + 4\,p1\cdot p2\,p3\cdot p4\,p5\cdot p6 - 4\,p1\cdot p6\,p2\cdot p4\,p3\cdot p5 + 4\,p1\cdot p4\,p2\cdot p6\,p3\cdot p5 + 4\,p1\cdot p5\,p2\cdot p4\,p3\cdot p6 - 4\,p1\cdot p4\,p2\cdot p5\,p3\cdot p6 + 4\,p1\cdot p6\,p2\cdot p3\,p4\cdot p5 - 4\,p1\cdot p3\,p2\cdot p6\,p4\cdot p5 + 4\,p1\cdot p2\,p3\cdot p6\,p4\cdot p5 - 4\,p1\cdot p5\,p2\cdot p3\,p4\cdot p6 + 4\,p1\cdot p3\,p2\cdot p5\,p4\cdot p6 - 4\,p1\cdot p2\,p3\cdot p5\,p4\cdot p6 + 4\,p1\cdot p4\,p2\cdot p3\,p5\cdot p6 - 4\,p1\cdot p3\,p2\cdot p4\,p5\cdot p6$

The Method of Region Expansion in NRQCD

- Expanding the relative momentum q Before or After loop integration?
  - Generally, we expand q after performing the loop integration, and then project the S-, P- or D-waves for the quarkonium.
  - We can also expand q before performing the loop integration, as long as only the hard region concerned, according to the method of region expansion [Benek, Smirnov (1998)].

- Hard region
  - If the NRQCD factorization is valid, it will be safe to use the method of region expand to compute the short-distance coefficients, which correspond to the hard region.

Passarino-Veltman Reduction

- Generic one-loop integral

$$\mathcal{T}^{\mu_1 \cdots \mu_p} \equiv \frac{(2\pi\mu)^{4-d}}{i\pi^2} \int d^d k \frac{k^{\mu_1} \ldots k^{\mu_p}}{D_0 D_1 D_2 \cdots D_{n-1}}$$



$$D_i = (k + r_i)^2 - m_i^2 + i\varepsilon$$

$$r_i = \sum_{k=1}^{i} p_k, \qquad i = 1, \cdots, n-1$$

$$r_0 = \sum_{k=1}^{n} p_k = 0$$

$$r_{ij} = r_i - r_j$$

Passarino-Veltman Reduction

- Generic n-Point Tensor integrals

$$B^{\mu}(r_{10}^2, m_0^2, m_1^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk\; k^{\mu}\prod_{i=0}^{1}\frac{1}{(k+r_i)^2-m_i^2}$$

$$B^{\mu\nu}(r_{10}^2, m_0^2, m_1^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk\; k^{\mu}k^{\nu}\prod_{i=0}^{1}\frac{1}{(k+r_i)^2-m_i^2}$$

$$C^{\mu}(r_{10}^2, r_{12}^2, r_{20}^2, m_0^2, m_1^2, m_2^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk\; k^{\mu}\prod_{i=0}^{2}\frac{1}{(k+r_i)^2-m_i^2}$$

$$C^{\mu\nu}(r_{10}^2, r_{12}^2, r_{20}^2, m_0^2, m_1^2, m_2^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk\; k^{\mu}k^{\nu}\prod_{i=0}^{2}\frac{1}{(k+r_i)^2-m_i^2}$$

$$C^{\mu\nu\rho}(r_{10}^2, r_{12}^2, r_{20}^2, m_0^2, m_1^2, m_2^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk\; k^{\mu}k^{\nu}k^{\rho}\prod_{i=0}^{2}\frac{1}{(k+r_i)^2-m_i^2}$$

$$D^{\mu}(r_{10}^2, r_{12}^2, r_{23}^2, r_{30}^2, r_{20}^2, r_{13}^2, m_0^2, m_1^2, m_2^2, m_3^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk\; k^{\mu}\prod_{i=0}^{3}\frac{1}{(k+r_i)^2-m_i^2}$$

$$D^{\mu\nu}(r_{10}^2, r_{12}^2, r_{23}^2, r_{30}^2, r_{20}^2, r_{13}^2, m_0^2, m_1^2, m_2^2, m_3^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk\; k^{\mu}k^{\nu}\prod_{i=0}^{3}\frac{1}{(k+r_i)^2-m_i^2}$$

$$D^{\mu\nu\rho}(r_{10}^2, r_{12}^2, r_{23}^2, r_{30}^2, r_{20}^2, r_{13}^2, m_0^2, m_1^2, m_2^2, m_3^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk\; k^{\mu}k^{\nu}k^{\rho}\prod_{i=0}^{3}\frac{1}{(k+r_i)^2-m_i^2}$$

$$D^{\mu\nu\rho\sigma}(r_{10}^2, r_{12}^2, r_{23}^2, r_{30}^2, r_{20}^2, r_{13}^2, m_0^2, m_1^2, m_2^2, m_3^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk\; k^{\mu}k^{\nu}k^{\rho}k^{\sigma}\prod_{i=0}^{3}\frac{1}{(k+r_i)^2-m_i^2}$$

Passarino-Veltman Reduction

- Generic n-Point Scalar integrals

$$A_0(m_0^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk \frac{1}{k^2-m_0^2}$$

$$B_0(r_{10}^2, m_0^2, m_1^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk \prod_{i=0}^{1}\frac{1}{(k+r_i)^2-m_i^2}$$

$$C_0(r_{10}^2, r_{12}^2, r_{20}^2, m_0^2, m_1^2, m_2^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk \prod_{i=0}^{2}\frac{1}{(k+r_i)^2-m_i^2}$$

$$D_0(r_{10}^2, r_{12}^2, r_{23}^2, r_{30}^2, r_{20}^2, r_{13}^2, m_0^2, m_1^2, m_2^2, m_3^2) = \frac{(2\pi\mu)^{4-d}}{i\pi^2}\int d^dk \prod_{i=0}^{3}\frac{1}{(k+r_i)^2-m_i^2}$$

- Pa-Ve Reduction

$$\begin{aligned} D^{\mu\nu\rho\sigma} &= (g^{\mu\nu}g^{\rho\sigma} + g^{\mu\rho}g^{\nu\sigma} + g^{\mu\sigma}g^{\nu\rho})D_{0000} \\ &= +\sum_{i,j}^{3}(g^{\mu\nu}r_i^\rho r_j^\sigma + g^{\nu\rho}r_i^\mu r_j^\sigma + g^{\mu\rho}r_i^\nu r_j^\sigma + g^{\mu\sigma}r_i^\nu r_j^\rho + g^{\nu\sigma}r_i^\mu r_j^\rho + g^{\rho\sigma}r_i^\mu r_j^\nu)D_{00ij} \\ &= +\sum_{i,j,k,l=1}^{3} r_\mu^i rj\nu r_k^\rho r_l^\sigma D_{ijkl} \end{aligned}$$

Passarino-Veltman Reduction

- Passarino-Veltman functions in FeynCalc

```
In[1]:= << HighEnergyPhysics`fc`

In[2]:= PaVe[0, 0, 0, 0, {r10^2, r12^2, r23^2, r30^2, r20^2, r13^2}, {m0^2, m1^2, m2^2, m3^2}]
```

Out[2]= D_{0000}

```
In[3]:= PaVe[0, 1, 2, 3, {r10^2, r12^2, r23^2, r30^2, r20^2, r13^2}, {m0^2, m1^2, m2^2, m3^2}]
```

Out[3]= D_{0123}

```
In[4]:= PaVe[0, 0, 0, 0, {1, 2, 3, 4, 5, 6}, {1, 1, 1, 1}] // PaVeReduce
```

Out[4]=

$$-\frac{135\,C_0(1,2,5,1,1,1)}{2401}-\frac{5751\,C_0(1,4,6,1,1,1)}{192\,080}-\frac{88\,691\,C_0(2,3,6,1,1,1)}{2\,650\,704}-\frac{5755\,C_0(3,4,5,1,1,1)}{633\,864}+\frac{1587\,D_0(1,2,3,4,5,6,1,1,1,1)}{38\,416}+\frac{51\,B_0(1,1,1)}{1960}+\frac{907\,B_0(2,1,1)}{27\,048}+\frac{1025\,B_0(3,1,1)}{99\,176}+\frac{347\,B_0(4,1,1)}{32\,340}-\frac{50\,B_0(5,1,1)}{1617}-\frac{181\,B_0(6,1,1)}{22\,540}+\frac{5}{72}$$

- Break-down for Gram Determinant = 0
  - Happens when we expand the relative momentum q before loop integration, due to taking the derivative over q.

Integrate-By-Part (IBP) Reduction

- Integrate By Part (IBP) [JHEP **0810**, 107 (2008)]

$$F(a_1, \cdots, a_n) = \int \cdots \int \frac{d^d k_1 \cdots d^d k_h}{E_1^{a_1} \cdots E_n^{a_n}}$$

where k_i, $i = 1, \cdots, h$, are loop momenta and the denominators E_r are either quadratic or linear with respect to the loop momenta k_i of the graph. Irreducible polynomials in the numerator can be represented as denominators raised to negative powers.

  - Basic idea:

$$\int \cdots \int d^d k_1 d^d k_2 \cdots \frac{\partial}{\partial k_i} \left[\frac{p_j}{E_1^{a_1} \cdots E_n^{a_n}} \right] = 0$$

  - List of equations:

$$\sum \alpha_i F(a_1 + b_{i,1}, \cdots, a_n + b_{i,n}) = 0$$

  - Master Integrals

- Propagators should be independent!

Partial Fraction on Propagators

- Apart in Mathematica

In[1]:= $\text{Apart}\left[\frac{1}{(x-a)\ (x-b)\ (x-c)}\right]$

Out[1]= $-\frac{1}{(a-b)(b-c)(x-b)}-\frac{1}{(a-c)(c-b)(x-c)}+\frac{1}{(a-b)(a-c)(x-a)}$

In[2]:= $\text{Apart}\left[\frac{1}{(x+a)\ (y+2\,b)\ (3\,x+4\,y)}\right]$

Out[2]= $\frac{1}{(a+x)(3\,x-8\,b)(2\,b+y)}-\frac{4}{(a+x)(3\,x-8\,b)(3\,x+4\,y)}$

- Do it with $Apart function

In[1]:= << CalcExt`Apart`

In[2]:= $\$\text{Apart}\left[\frac{1}{(x-a)\ (x-b)\ (x-c)},\ \{x\}\right]$

Out[2]= $-\frac{\left\|\frac{1}{a-x}\right\|}{(a-b)(a-c)}+\frac{\left\|\frac{1}{b-x}\right\|}{(a-b)(b-c)}+\frac{\left\|\frac{1}{c-x}\right\|}{(a-c)(c-b)}$

In[3]:= $\$\text{Apart}\left[\frac{1}{(x+a)\ (y+2\,b)\ (3\,x+4\,y)},\ \{x,\ y\}\right]$

Out[3]= $-\frac{\left\|\frac{1}{(a+x)(2\,b+y)}\right\|}{3\,a+8\,b}+\frac{4\left\|\frac{1}{(a+x)(3\,x+4\,y)}\right\|}{3\,a+8\,b}+\frac{3\left\|\frac{1}{(2\,b+y)(3\,x+4\,y)}\right\|}{3\,a+8\,b}$

Partial Fraction on Propagators

- What does Linear Independent mean?
  - V_x : vector space spanned by n linear-independent vectors: $\{x_i\}$ over the coefficient field $\mathcal{F}$.
  - $V_x^* \equiv V_x \oplus \mathcal{F}$
  - $\{e_i = v_i + f_i\}$ are linear independent if and only if $\{v_i\}$ are linear indepent

$$\sum f_i \, e_i = f \quad (\text{not all } f_i = 0)$$

  - Product Operator: inherited from n-variate polynormial
- What does $Apart do? [Comput.Phys.Commun.183,2158(2012)]

$$\prod_{i=1}^{N} e_i^{n_i} = \sum_j f_j \prod_{i=1}^{N_j} e_{k_{ji}}^{n_{ji}}$$

where $0 < k_{ji} < N$, and $\{e_{k_{ji}}\}$ are lienar independent.

Partial Fraction on Propagators

- A simply physical example:

$$\exp = \frac{(k\cdot p_1)\,(k\cdot p_2)}{k^2[(k+p_1)^2-m^2][(k+p_2)-m^2]^2}$$

Taking k^2, $k\cdot p_1$ and $k\cdot p_2$ as indepdent vectors.

- Partial Fraction on it:

```
In[1]:= << CalcExt`Apart`

In[2]:= << HighEnergyPhysics`fc`

In[3]:= exp = SP[k, p1] SP[k, p2] / (SP[k] (SP[k + p1] - m^2) (SP[k + p2] - m^2)^2) // FCI // ScalarProductExpand;

In[4]:= xs = FCI /@ {SP[k], SP[k, p1], SP[k, p2]}
```

Out[4]= $\{k^2,\ k\cdot \mathrm{p1},\ k\cdot \mathrm{p2}\}$

```
In[5]:= $Apart[exp, xs]
```

Out[5]=

$$-\frac{1}{4}\left\|\frac{1}{(m^2-k^2-2k\cdot \mathrm{p2}-\mathrm{p2}^2)^2}\right\|+\frac{1}{4}(m^2-\mathrm{p1}^2)\left\|\frac{1}{(m^2-k^2-2k\cdot \mathrm{p1}-\mathrm{p1}^2)(m^2-k^2-2k\cdot \mathrm{p2}-\mathrm{p2}^2)^2}\right\|+\frac{1}{4}(m^2-\mathrm{p2}^2)\left\|\frac{1}{k^2(-m^2+k^2+2k\cdot \mathrm{p2}+\mathrm{p2}^2)^2}\right\|+$$

$$\frac{1}{2}\left\|\frac{k\cdot \mathrm{p2}}{(m^2-k^2-2k\cdot \mathrm{p1}-\mathrm{p1}^2)(-m^2+k^2+2k\cdot \mathrm{p2}+\mathrm{p2}^2)^2}\right\|+\frac{1}{4}(m^2-\mathrm{p1}^2)(m^2-\mathrm{p2}^2)\left\|\frac{1}{k^2(-m^2+k^2+2k\cdot \mathrm{p1}+\mathrm{p1}^2)(-m^2+k^2+2k\cdot \mathrm{p2}+\mathrm{p2}^2)^2}\right\|+$$

$$\frac{1}{4}\left\|\frac{1}{k^2(-m^2+k^2+2k\cdot \mathrm{p2}+\mathrm{p2}^2)}\right\|+\frac{1}{4}(m^2-\mathrm{p1}^2)\left\|\frac{1}{k^2(-m^2+k^2+2k\cdot \mathrm{p1}+\mathrm{p1}^2)(-m^2+k^2+2k\cdot \mathrm{p2}+\mathrm{p2}^2)}\right\|$$

Integrate-By-Part (IBP) Reduction

- Public codes performing IBP reduction
  - AIR – Maple JHEP **0407** (2004) 046
  - FIRE – Mathematica/C++ JHEP **0810**, 107 (2008)
  - Reduze – C++ Comput. Phys. Commun. 181, 1293 (2010)
  - LiteRed – Mathematica arXiv:1212.2685 (2012)
  - Many other private codes
- FIRE – Feynman Integral Reduction

```
In[5]:= $Apart[exp, xs]
```

Out[5]=

$$-\frac{1}{4}\left\|\frac{1}{(m^2-k^2-2k\cdot p2-p2^2)^2}\right\|+\frac{1}{4}(m^2-p1^2)\left\|\frac{1}{(m^2-k^2-2k\cdot p1-p1^2)(m^2-k^2-2k\cdot p2-p2^2)^2}\right\|+\frac{1}{4}(m^2-p2^2)\left\|\frac{1}{k^2(-m^2+k^2+2k\cdot p2+p2^2)^2}\right\|+\frac{1}{2}\left\|\frac{k\cdot p2}{(m^2-k^2-2k\cdot p1-p1^2)(-m^2+k^2+2k\cdot p2+p2^2)^2}\right\|+\frac{1}{4}(m^2-p1^2)(m^2-p2^2)\left\|\frac{1}{k^2(-m^2+k^2+2k\cdot p1+p1^2)(-m^2+k^2+2k\cdot p2+p2^2)^2}\right\|+\frac{1}{4}\left\|\frac{1}{k^2(-m^2+k^2+2k\cdot p2+p2^2)}\right\|+\frac{1}{4}(m^2-p1^2)\left\|\frac{1}{k^2(-m^2+k^2+2k\cdot p1+p1^2)(-m^2+k^2+2k\cdot p2+p2^2)}\right\|$$

Integrate-By-Part (IBP) Reduction

- FIRE – Feynman Integral Reduction

$$F[l,m,n] = \int \frac{d^4k}{(2\pi)^4} \frac{(k\cdot p_2)^{-l}}{(m^2-k^2-2k\cdot p_1-p_1^2)^m(-m^2+k^2+2k\cdot p_2+p_2^2)^n}$$

- Basic usage of FIRE

```
In[1]:= << HighEnergyPhysics`fc`

In[2]:= << FIRE`

In[3]:= Replacement = {p1^2 → m^2, p2^2 → m^2, p1 p2 → SP[p1, p2]};
        Internal = {k};
        External = {p1, p2};
        Propagators = {k p2, -2 k p1 - k^2 + m^2 - p1^2, 2 k p2 + k^2 - m^2 + p2^2};
        PrepareIBP[];
        startinglist = {IBP[k, k], IBP[k, p1], IBP[k, p2]} /. Replacement;
        Prepare[];

In[10]:= Burn[];

In[11]:= F[{-1, 1, 2}]
```

Out[11]= $\frac{(d-2)\,G(\{0,0,1\})}{8\,(m^2-\text{p1}\cdot\text{p2})} + \frac{(d-2)\,G(\{0,1,0\})}{8\,(m^2-\text{p1}\cdot\text{p2})} + \frac{1}{4}(4-d)\,G(\{0,1,1\})$

Final Result

- After Partial Fraction and IBP Reduction:

$$\int \frac{d^4k}{(2\pi)^4} \frac{(k\cdot p_1)\,(k\cdot p_2)}{k^2[(k+p_1)^2-m^2][(k+p_2)-m^2]^2}$$

Assuming $p_1^2 = p_2^2 = m^2$:

Out[27]=

$$\frac{(D-2)\left\|\frac{1}{-2\,k\cdot \mathrm{p1}-k^2}\right\|}{16\,(m^2-\mathrm{p1}\cdot\mathrm{p2})} + \frac{(D-2)\left\|\frac{1}{2\,k\cdot \mathrm{p2}+k^2}\right\|}{16\,(m^2-\mathrm{p1}\cdot\mathrm{p2})} + \frac{1}{8}(4-D)\left\|\frac{1}{(-2\,k\cdot\mathrm{p1}-k^2)\,(2\,k\cdot\mathrm{p2}+k^2)}\right\| + \frac{1}{4}\left\|\frac{1}{k^2\,(2\,k\cdot\mathrm{p2}+k^2)}\right\|$$

where $\left\|\cdots\right\|$ is defined as:

$$\left\|\exp\right\| = \int \frac{d^4k}{(2\pi)^4}\exp$$

Using Apart

In[2]:= `<< CalcExt`Apart``

In[3]:= $\text{exp} = \dfrac{\text{SP[k, p1] SP[k, p2]}}{\text{SP[k]} \left(\text{SP[k + p1]} - \text{m}^2\right) \left(\text{SP[k + p2]} - \text{m}^2\right)^2}$ `// FCI // ScalarProductExpand;`

In[4]:= `xs = FCI /@ {SP[k], SP[k, p1], SP[k, p2]}`

Out[4]= $\{k^2, k\cdot \text{p1}, k\cdot \text{p2}\}$

In[5]:= `ApartReasult = $Apart[exp, xs]`

Out[5]= $$-\frac{1}{4}\left\|\frac{1}{(m^2-k^2-2k\cdot \text{p2}-\text{p2}^2)^2}\right\| + \frac{1}{4}(m^2-\text{p1}^2)\left\|\frac{1}{(m^2-k^2-2k\cdot\text{p1}-\text{p1}^2)(m^2-k^2-2k\cdot\text{p2}-\text{p2}^2)^2}\right\| + \frac{1}{4}(m^2-\text{p2}^2)\left\|\frac{1}{k^2(-m^2+k^2+2k\cdot\text{p2}+\text{p2}^2)^2}\right\| +$$
$$\frac{1}{2}\left\|\frac{k\cdot\text{p2}}{(m^2-k^2-2k\cdot\text{p1}-\text{p1}^2)(-m^2+k^2+2k\cdot\text{p2}+\text{p2}^2)^2}\right\| + \frac{1}{4}(m^2-\text{p1}^2)(m^2-\text{p2}^2)\left\|\frac{1}{k^2(-m^2+k^2+2k\cdot\text{p1}+\text{p1}^2)(-m^2+k^2+2k\cdot\text{p2}+\text{p2}^2)^2}\right\| +$$
$$\frac{1}{4}\left\|\frac{1}{k^2(-m^2+k^2+2k\cdot\text{p2}+\text{p2}^2)}\right\| + \frac{1}{4}(m^2-\text{p1}^2)\left\|\frac{1}{k^2(-m^2+k^2+2k\cdot\text{p1}+\text{p1}^2)(-m^2+k^2+2k\cdot\text{p2}+\text{p2}^2)}\right\|$$

In[6]:= `ApartReasult = $ApartComplete[ApartReasult, xs];`

Generate All Propagtors

In[7]:= `PropagatorsList = Union @@ (($ApartUnion[#] /. $ApartIR[_, _, x_, _] :> x) & /@ Flatten[{ApartReasult}])`

Out[7]= $$\begin{pmatrix} k^2 & -m^2+k^2+2k\cdot\text{p1}+\text{p1}^2 & -m^2+k^2+2k\cdot\text{p2}+\text{p2}^2 \\ k^2 & -m^2+k^2+2k\cdot\text{p2}+\text{p2}^2 & k\cdot\text{p1} \\ k\cdot\text{p2} & m^2-k^2-2k\cdot\text{p1}-\text{p1}^2 & -m^2+k^2+2k\cdot\text{p2}+\text{p2}^2 \\ m^2-k^2-2k\cdot\text{p1}-\text{p1}^2 & m^2-k^2-2k\cdot\text{p2}-\text{p2}^2 & k^2 \\ m^2-k^2-2k\cdot\text{p2}-\text{p2}^2 & k^2 & k\cdot\text{p1} \end{pmatrix}$$

In[8]:= `For[i = 1, i ≤ Length[PropagatorsList], i++, PNS[PropagatorsList[[i]]] = i];`

In[9]:= `SPropagatorsList = PropagatorsList /. {FCI@SP[k] → k^2, FCI@SP[k, p1] → k p1, FCI@SP[k, p2] → k p2};`

Setup FIRE

```
In[10]:= << FIRE`
```

Init Parameters

```
In[11]:= Replacement = {p1^2 → m^2, p2^2 → m^2, p1 p2 → SP[p1, p2]} // FCI;
         d = D;
```

Prepare

```
In[13]:= For[$i = 1, $i ≤ Length[SPropagatorsList], $i ++,
           ClearIBP[];
           Internal = {k};
           External = {p1, p2};
           Propagators = {k p2, -2 k p1 - k^2 + m^2 - p1^2, 2 k p2 + k^2 - m^2 + p2^2};
           PrepareIBP[];
           startinglist = {IBP[k, k], IBP[k, p1], IBP[k, p2]} /. Replacement;
           Prepare[$i];
          ];
         ClearIBP[];
```

Burn

```
In[15]:= Burn[];
```

Results

```
In[16]:= Clear[IntF];
FireReduce[exp_] := exp /. $ApartIR[_, _, x_, n_] ↦ F[PNS[x], -n] /. G[pn_, ns_] ↦ IntF[PropagatorsList[[pn]], -ns];
IntF[x_List, n_List] := IntF[Times @@ x^n, q1];
IntF /: MakeBoxes[IntF[exp_, ___], TraditionalForm] := RowBox[{"‖", MakeBoxes[exp, TraditionalForm], "‖"}];
IntFCollect[exp_] := Collect[exp, _IntF];

In[21]:= FireReduce[ApartReasult] /. {FCI@SP[p1] → m^2, FCI@SP[p2] → m^2}
```

Out[21]=

$$\frac{1}{2}\left(\frac{(D-2)\left\|\frac{1}{-2k\cdot p1-k^2}\right\|}{8(m^2-p1\cdot p2)}+\frac{(D-2)\left\|\frac{1}{2k\cdot p2+k^2}\right\|}{8(m^2-p1\cdot p2)}+\frac{1}{4}(4-D)\left\|\frac{1}{(-2k\cdot p1-k^2)(2k\cdot p2+k^2)}\right\|\right)+\frac{1}{4}\left\|\frac{1}{k^2(2k\cdot p2+k^2)}\right\|$$

```
In[22]:= IntFCollect[%]
```

Out[22]=

$$\frac{(D-2)\left\|\frac{1}{-2k\cdot p1-k^2}\right\|}{16(m^2-p1\cdot p2)}+\frac{(D-2)\left\|\frac{1}{2k\cdot p2+k^2}\right\|}{16(m^2-p1\cdot p2)}+\frac{1}{8}(4-D)\left\|\frac{1}{(-2k\cdot p1-k^2)(2k\cdot p2+k^2)}\right\|+\frac{1}{4}\left\|\frac{1}{k^2(2k\cdot p2+k^2)}\right\|$$

Summary

- Using FeynArts for Feyman diagrams and amplitudes generation.
- Using FeynCalc/FormLink for Dirac- and Color-algebra simplification.
- Using $Apart and FIRE for partial fraction and Integrate-By-Part reduction respectively.
- Solving Master Integrals by any other mean and substituting them to get the final result.
- Post-processing the final result in last previous step.

Thank You!

TID

```
In[1]:= << HighEnergyPhysics`fc`
```

```
In[2]:= << CalcExt`TID`
```

```
In[3]:= FVD[k, μ] f[SP[p1], SP[p2], SP[p, p2]] // FCI
```

Out[3]= $k^{\mu}\, f(p1^2, p2^2, p\cdot p2)$

```
In[4]:= $TID[FVD[k, μ] f[SP[p1], SP[p2], SP[p, p2]] // FCI, k, {p1, p2}]
```

Out[4]= $$\left(\frac{p2^{\mu}\, k\cdot p1\; p1\cdot p2 - p1^2\, p2^{\mu}\, k\cdot p2}{p1\cdot p2^2 - p1^2\, p2^2} + \frac{p1^{\mu}\, k\cdot p2\; p1\cdot p2 - p2^2\, p1^{\mu}\, k\cdot p1}{p1\cdot p2^2 - p1^2\, p2^2}\right) f(p1^2, p2^2, p\cdot p2)$$

```
In[5]:= FVD[k, μ] FVD[k, ν] f[SP[p1], SP[p2], SP[p, p2]] // FCI
```

Out[5]= $k^{\mu}\, k^{\nu}\, f(p1^2, p2^2, p\cdot p2)$

```
In[6]:= $TID[FVD[k, μ] FVD[k, ν] f[SP[p1], SP[p2], SP[p, p2]] // FCI, k, {p1, p2}]
```

Out[6]= $$f(p1^2, p2^2, p\cdot p2)\Bigg(\frac{g^{\mu\nu}\left(p2^2\, k\cdot p1^2 - 2\, k\cdot p2\; p1\cdot p2\; k\cdot p1 + k^2\, p1\cdot p2^2 + k\cdot p2^2\, p1^2 - k^2\, p1^2\, p2^2\right)}{(D-2)\left(p1\cdot p2^2 - p1^2\, p2^2\right)} + \frac{1}{(D-2)\left(p1\cdot p2^2 - p1^2\, p2^2\right)^2}\, p2^{\mu}\, p2^{\nu}\big(D\, p1\cdot p2^2\, k\cdot p1^2 - 2\, p1\cdot p2^2\, k\cdot p1^2 +$$
$$p1^2\, p2^2\, k\cdot p1^2 - 2\, D\, k\cdot p2\; p1^2\; p1\cdot p2\; k\cdot p1 + 2\, k\cdot p2\; p1^2\; p1\cdot p2\; k\cdot p1 + D\, k\cdot p2^2\, {p1^2}^2 - k\cdot p2^2\, {p1^2}^2 + k^2\, p1^2\, p1\cdot p2^2 - k^2\, {p1^2}^2\, p2^2\big) +$$
$$\frac{1}{(D-2)\left(p1\cdot p2^2 - p1^2\, p2^2\right)^2}\, p2^{\mu}\, p1^{\nu}\big(-k^2\, p1\cdot p2^3 + D\, k\cdot p1\; k\cdot p2\; p1\cdot p2^2 - D\, k\cdot p2^2\, p1^2\; p1\cdot p2 + k\cdot p2^2\, p1^2\; p1\cdot p2 -$$
$$D\, k\cdot p1^2\, p2^2\; p1\cdot p2 + k\cdot p1^2\, p2^2\; p1\cdot p2 + k^2\, p1^2\, p2^2\; p1\cdot p2 + D\, k\cdot p1\; k\cdot p2\; p1^2\, p2^2 - 2\, k\cdot p1\; k\cdot p2\; p1^2\, p2^2\big) +$$
$$\frac{1}{(D-2)\left(p1\cdot p2^2 - p1^2\, p2^2\right)^2}\, p1^{\mu}\, p2^{\nu}\big(-k^2\, p1\cdot p2^3 + D\, k\cdot p1\; k\cdot p2\; p1\cdot p2^2 - D\, k\cdot p2^2\, p1^2\; p1\cdot p2 + k\cdot p2^2\, p1^2\; p1\cdot p2 -$$
$$D\, k\cdot p1^2\, p2^2\; p1\cdot p2 + k\cdot p1^2\, p2^2\; p1\cdot p2 + k^2\, p1^2\, p2^2\; p1\cdot p2 + D\, k\cdot p1\; k\cdot p2\; p1^2\, p2^2 - 2\, k\cdot p1\; k\cdot p2\; p1^2\, p2^2\big) +$$
$$\frac{1}{(D-2)\left(p1\cdot p2^2 - p1^2\, p2^2\right)^2}\, p1^{\mu}\, p1^{\nu}\big(D\, p1\cdot p2^2\, k\cdot p2^2 - 2\, p1\cdot p2^2\, k\cdot p2^2 + p1^2\, p2^2\, k\cdot p2^2 - 2\, D\, k\cdot p1\; p1\cdot p2\; p2^2\, k\cdot p2 +$$
$$2\, k\cdot p1\; p1\cdot p2\; p2^2\, k\cdot p2 + D\, k\cdot p1^2\, {p2^2}^2 - k\cdot p1^2\, {p2^2}^2 - k^2\, p1^2\, {p2^2}^2 + k^2\, p1\cdot p2^2\, p2^2\big)\Bigg)$$